
 
 
 
 
 
 
 
 
 

Unul din principalele atribute ale datelor este structura sau modul de 
organizare care le caracterizează. Structurile rezultă prin gruparea într-un anumit 
mod a colecţiilor de date primitive. Exemple de structuri de date sunt: tablourile, 
mulţimile, înregistrările. În continuare vor fi prezentate noi tipuri de organizări, şi 
anume: listele simplu şi dublu înlănţuite şi cazurile particulare cunoscute sub numele 
de stive şi cozi. 
 
 

2.1  Noţiuni introductive 
 
 
Un tip de structură de date poate fi gândit logic, făcând abstracţie de modul 

particular în care datele sunt reprezentate în memoria calculatorului. El se reprezintă 
particularizat prin diverse implementări din limbajele de programare. 

Specificarea unui anume tip de organizare pentru datele procesate de un 
program presupune definirea mulţimii de valori permise pe care o variabilă sau 
parametru aparţinând acestui tip de date le poate avea, precum şi a mulţimii de 
operaţii care pot fi efectuate cu astfel de date. Eficienţa unei scheme de prelucrare a 
datelor este determinată în mare măsură de tipurile particulare de date utilizate.  

Limbajul de programare Pascal oferă suficientă flexibilitate pentru 
reprezentarea unor tipuri complexe de date, dar controlul corectitudinii modului în 
care sunt utilizate revine programatorului. Structurile de date Pascal pentru 
reprezentarea colecţiilor de date pot fi clasificate în structuri statice şi dinamice. 
Structurile statice cuprind: array, set şi file pentru reprezentarea datelor de acelaşi tip 
(omogene) şi record, pentru reprezentarea datelor eterogene. Structurile dinamice se 
construiesc pe baza tipurilor de date Pascal referinţă legată şi referinţă liberă 
(desemnată prin tipul predefinit pointer). 

Definirea unei date reprezentată printr-o structură statică presupune alocarea 
unui spaţiu de memorie de dimensiune invariabilă pe durata evoluţiei programului, în 
timp ce dimensiunea unei date reprezentată prin intermediul unei structuri dinamice 
poate fi modificată pe durata execuţiei. 

STRUCTURA DE LISTĂ 
ÎN LIMBAJUL PASCAL 

 Pentru anumite clase de probleme, structurile statice de date nu numai că nu 
sunt suficiente, dar se dovedesc chiar imposibil de folosit, datorită limitărilor la care 
sunt supuse. În primul rând, spaţiul de memorie aferent unor astfel de date se 
defineşte şi se rezervă în momentul compilării programului (rezervare statică), la o 
dimensiune maximă (cuprinzătoare). Acest spaţiu nu poate fi disponibilizat şi nici 
împărţit cu alte date, chiar dacă nu este în întregime utilizat în anumite momente ale 
execuţiei programului. În al doilea rând, componentele structurilor statice ocupă 
locuri prestabilite în spaţiul rezervat, determinate de relaţia de ordonare specifică 
fiecărei structuri. În al treilea rând, limbajul defineşte operaţiile admise cu valorile 
componentelor, potrivit tipului de bază al structurii, astfel încât numărul maxim şi 
ordinea componentelor structurii nu pot fi modificate. 
 În aceste condiţii, structurile statice sunt dificil de utilizat în rezolvarea 
problemelor care prelucrează mulţimi de date pentru care numărul şi ordinea 
componentelor se modifică frecvent în timpul execuţiei programului. Pentru astfel de 
situaţii, există posibilitatea utilizării datelor de tip dinamic, cărora li se pot aloca şi 
elibera zone de memorie pe parcursul execuţiei programului. 
 
 

 
2.2  Structura de listă 

 
 

Organizarea de tip listă corespunde unei structurări lineare a datelor, la 
nivelul fiecărei componente dispunându-se de informaţie suficientă pentru 
identificarea următoarei componente a colecţiei. Datele dintr-o colecţie astfel 
structurată sunt referite de obicei prin termenii de noduri, celule, componente etc. 

Reprezentarea unei liste în limbajul Pascal poate fi realizată prin intermediul 
structurii de date array, ordinea componentelor fiind dată de ordinea pe domeniul 
valorilor corespunzătoare indexării şi, în consecinţă, următoarea componentă este 
implicit specificată. Memorarea unei mulţimi de date {d1, d2,…, dn} prin intermediul 
unei structuri statice poate fi realizată în limbajul Pascal utilizând un masiv 
unidimensional.  

Definirea tipului de date Pascal pentru memorarea ca listă a datelor {d1, 
d2,…, dn} este: 

const max=500; 
type lst=array[1..max] of tip_informatie; 

unde n≤max şi tip_informatie este numele tipului de date Pascal utilizat pentru 
memorarea fiecărei date din mulţimea {d1, d2,…, dn}. Dacă lista este variabila de  
tip lst utilizată pentru memorarea colecţiei {d1, d2,…, dn}, atunci data di este 
memorată în componenta lista[i],  ni1 ≤≤ . 

Dezavantajele majore în utilizarea reprezentării statice rezidă în volumul de 
calcule necesare efectuării operaţiilor de inserţie/eliminare de noduri şi în necesitatea 
păstrării unei zone de memorie alocată, indiferent de lungimea efectivă a listei. 
Aceste dezavantaje sunt eliminate prin opţiunea de utilizare a structurilor dinamice. 
Componentele unei liste dinamice sunt eterogene, fiecare nod conţinând o parte de 
informaţie şi câmpuri de legătură care permit identificarea celulelor vecine. 
Câmpurile de legătură sunt reprezentate de date de tip referinţă (adresă). În cazul 
listelor cu un singur câmp de legătură (simplu înlănţuite),  valoarea câmpului indică 
adresa nodului următor, în timp ce, în cazul listelor cu dublă legătură (dublu 
înlănţuite), valorile memorate în câmpurile de legătură sunt adresele componentelor 
care preced şi, respectiv, urmează celulei. În ambele situaţii, câmpul de legătură 
pentru indicarea celulei următoare corespunzător ultimei componente a listei are 
valoarea nil în cazul listelor “deschise” (lineare) şi respectiv indică adresa primei 
componente din listă în cazul listelor “închise” (circulare).  

Se presupune că mulţimea de date {d1, d2,…, dn} este memorată ca listă 
lineară. Convenţional, structura rezultată poate fi reprezentată grafic fie prin a) (listă 
simplu înlănţuită), fie prin b) (listă dublu înlănţuită), unde simbolul      

desemnează valoarea nil a câmpului de legătură (figura 2.1).  În cazul în care 
pentru memorarea datelor {d1, d2,…, dn} se doreşte utilizarea unei structuri de listă 
circulară, reprezentarea grafică este dată în figura 2.2. 

 
 dn dn-1 d1 d2

C 

… 

Fig. 2.1  Liste lineare

Declarările tipurilor de date Pascal pentru definirea structurilor de liste 

dinamice simplu şi, respectiv, dublu înlănţuite sunt: 
a) Listă simplu înlănţuită 
 
type lista=^nod; 
        nod=record 
                     inf:tip_informatie; 
                      leg: lista; 
                 end; 

b) Listă dublu înlănţuită 
 
type lista=^nod; 
        nod=record 
                     inf:tip_informatie; 
                      st,dr: lista; 
                 end; 

unde tip_informatie este numele tipului de date Pascal utilizat pentru memorarea 
fiecărei date din mulţimea{d1, d2,…, dn}. 

Fig. 2.2  Liste circulare

2.3 Operaţii primitive asupra listelor 
 
 

Accesul la informaţia stocată într-o variabilă de tip listă revine efectuând una 
sau mai multe dintre operaţiile primitive: regăsirea nodului (dacă există) care 
corespunde unei chei date (condiţie impusă asupra valorii câmpului de informaţie), 
inserarea unei noi componente în listă, eliminarea componentei (componentelor) cu 
proprietatea că valorile câmpurilor de informaţie satisfac o anumită cerinţă şi 
înlocuirea câmpului de informaţie corespunzător unei componente printr-o informaţie 
dată.  

Accesarea componentelor unei liste reprezentată printr-o structură statică 
poate fi realizată atât secvenţial, cât şi direct, utilizând valorile indicelui considerat 
pentru indexare, în timp ce accesarea componentelor unei liste dinamice se 
realizează, de regulă, numai secvenţial, începând cu prima componentă şi continuând 
cu următoarele, pe baza valorilor câmpurilor de legătură. Convenţional, numim cap 
al listei dinamice pointerul a cărui valoare este adresa primei componente a listei. 

 
1. Parcurgerea integrală a datelor memorate într-o listă. Se presupune că 

declarările de tip pentru definirea structurilor de liste menţionate anterior sunt 
globale, relativ la procedurile descrise în continuare. 

 



Structura de listă în limbajul Pascal 

a) Lista reprezentată prin structură statică (masiv unidimensional) 
 

procedure parcurgere1(var l:lst; n:word); 
var i:word; 
begin 
 for i:=1 to n do 
  prelucrare(l[i]); 
end; 
 
b) Lista reprezentată prin structură dinamică simplu înlănţuită 
 
procedure parcurgere2( var cap:lista); 
var p:lista; 
begin 
 p:=cap; 
   while(p<>nil) do 
  begin 
  prelucrare(p^.inf); 
  p:=p^.leg; 
  end; 
end; 
 
c)  Lista reprezentată prin structură dinamică dublu înlănţuită 
 
procedure parcurgere3( var cap:lista); 
var p:lista; 
begin 
 p:=cap; 
   while(p<>nil) do 
  begin 
  prelucrare(p^.inf); 
  p:=p^.dr; 
  end; 
end; 
 
2.  Regăsirea unei date d într-o colecţie memorată într-o listă 
În continuare se presupune că funcţia egal(a,b:tip_informatie):boolean 

returnează valoarea true dacă a şi b coincid, altfel returnează false. Funcţiile Pascal 
exista1, exista2 şi exista3 calculează valoarea true, dacă data d se află în colecţia 
memorată în lista dată ca argument, altfel calculează valoarea false. 

 

a) Lista reprezentată prin structură statică (masiv unidimensional) 
 

function exista1(l:lst; n:word; d:tip_informatie):boolean; 
var i:word; 
      c:boolean;  
begin 
 c:=false; 
   i:=1; 
 while(i<=n)and(not c) do 
  if egal(d,l[i]) then c:=true 
  else i:=i+1; 
 exista1:=c; 
end; 

b) Lista reprezentată prin structură dinamică simplu înlănţuită 
 
function exista2(cap:lista;d:tip_informatie):boolean; 
var p:lista; 

      c:boolean; 
begin 
 p:=cap; 
   c:=false; 
   while(p<>nil)and(not c) do 
  if egal(d,p^.inf) then c:=true 
  else p:=p^.leg; 
  exista2:=c; 
end; 

 
c) Lista reprezentată prin structură dinamică dublu înlănţuită 
 
function exista3(cap:lista;d:tip_informatie):boolean; 
var p:lista; 
      c:boolean; 
begin 
 p:=cap; 
   c:=false; 
   while(p<>nil)and(not c) do 
  if egal(d,p^.inf) then c:=true 
  else p:=p^.dr; 
  exista3:=c; 
end; 

 
3. Inserarea unei date, d, într-o listă 
Includerea unei noi componente într-o listă poate fi realizată, în funcţie de 

cerinţele problemei particulare, la începutul listei, după ultima componentă din listă, 
înaintea/după o componentă cu proprietatea că valoarea câmpului de infor- 
maţie îndeplineşte o anumită condiţie. Presupunem că funcţia condiţie 
((a,b:tip_informatie):boolean calculează valoarea true dacă a şi b verifică condiţia 
formulată pentru inserare, altfel calculează valoarea false. De asemenea, procedura 
atribuie(var destinatie:tip_informatie; sursa:tip_informatie) realizează copierea datei 
sursa în destinatie. 

Deoarece prin inserarea unei componente se poate ajunge la depăşirea 
spaţiului disponibil de memorie, este necesară verificarea prealabilă a posibilităţii 
realizării operaţiei (dacă se poate aloca spaţiu de memorie pentru componenta de 
inserat). În cazul listelor reprezentate prin structuri statice, această verificare va fi 
realizată prin compararea lungimii efective n a listei cu dimensiunea declarată max. 
În cazul listelor dinamice, este utilizată funcţia Pascal predefinită MaxAvail, care 
calculează lungimea maximă a zonelor nealocate contigue din memoria Heap. 
Valoarea returnată pentru parametrul test este true, dacă inserarea a fost posibilă, 
altfel este false. 

De exemplu,  inserarea la începutul listei decurge astfel: 
 
a) Lista reprezentată prin structură statică (masiv unidimensional) 

 
procedure inserare_la_inceput1(var l:lst; var n:word; 
d:tip_informatie;var test:boolean); 

var i:word; 
begin 
test:=true; 
if n=max then test:=false 
else 
 begin 
     for i:=n downto 1 do atribuie(l[i+1],l[i]); 
    atribuie(l[1],d); 
      n:=n+1; 
   end; 
end; 

 
b) Lista reprezentată prin structură dinamică simplu înlănţuită 
 
procedure inserare_la_inceput2(var 
cap:lista;d:tip_informatie;var test:boolean); 
var p:lista; 
 begin 
 test:=true; 
 if MaxAvail<sizeof(nod) then test:=false 
 else  
     begin 
     new(p); 
     atribuie(p^.inf,d); 
     p^.leg:=cap; 
     cap:=p; 
     end; 
 end; 

 
c) Lista reprezentată prin structură dinamică dublu înlănţuită 
 
procedure inserare_la_inceput3(var 
cap:lista;d:tip_informatie;var test:boolean); 
var p:lista; 
 begin 
 test:=true; 
 if MaxAvail<sizeof(nod) then test:=false 
 else  
     begin 
     new(p); 
     atribuie(p^.inf,d); 
     p^.dr:=cap;p^.st:=nil; 
     cap:=p; 
     end; 
 end; 

 
Procedurile de inserare la începutul listei pot fi folosite şi pentru crearea unei 

liste în care să fie memorată o colecţie de date {d1, d2,…, dn}. Procedura copiaza(var 
d:tip_informatie; var i:word) realizează copierea în d a datei di din colecţia 
considerată. Utilizând procedurile de inserare deja descrise, operaţia de creare a unei 
liste poate fi realizată astfel: 

 
a) Lista reprezentată prin structură statică (masiv unidimensional) 
 
procedure creeaza1(var l:lst; n:word; var test:boolean); 
var i,j:word; 

      d:tip_informatie; 
begin 
test:=true; 
if n>max then test:=false 
else  
  begin 
  i:=1; 
  while(i<=n) do  
     begin 
    copiaza(d,i); 
     inserare_la_inceput1(l,i,d,test) 
     end; 
  end; 
end; 

 
b) Lista reprezentată prin structură dinamică simplu înlănţuită 
 
procedure creeaza2(var cap:lista;n:word;var test:boolean); 
var i:word; 
 begin 
 i:=0; cap:=nil; 
 test:=true; 
 while(i<n)and(test) do 
           begin 
            i:=i+1;copiaza(d,i); 
           inserare_la_inceput2(cap,d,test); 
           end; 
end; 

 
4. Eliminarea (ştergerea) unei date, d, dintr-o listă 
Modificarea conţinutului unei liste prin eliminarea uneia sau mai multor 

componente poate fi descrisă secvenţial, astfel încât este suficient să dispunem de o 
procedură care realizează eliminarea unei singure componente. 

Criteriile de eliminare pot fi formulate diferit, cele mai uzuale fiind: prima 
componentă, ultima componentă, prima componentă care îndeplineşte o anumită 
condiţie, respectiv componenta ce precede/urmează prima componentă care 
îndeplineşte o condiţie dată. 

În aceste cazuri este necesară verificarea existenţei în lista considerată a 
componentei ce trebuie eliminată. Verificarea asigură şi testarea faptului că lista 
prelucrată este vidă sau nu. Analog operaţiilor de inserare, în cadrul procedurilor 
următoare, parametrul test returnează valoarea true, dacă eliminarea este efectivă, 
altfel returnează false. 

De exemplu, eliminarea primei componente dintr-o listă are loc astfel: 
 
a) Lista reprezentată prin structură statică (masiv unidimensional) 

 
procedure elimina_prima1(var l:lst; var n:word; var 
d:tip_informatie;var test:boolean); 
var i:word; 
begin 
test:=true; 
if n=0 then test:=false 
else 
 begin 
  atribuie(d,l[1]); 

  for i:=1 to n-1 do atribuie(l[i],l[i+1]); 
    n:=n-1; 
   end; 
end; 
 
b) Lista reprezentată prin structură dinamică simplu înlănţuită 
 
procedure elimina_prima2(var cap:lista;var d:tip_informatie;var 
test:boolean); 
var p:lista; 
 begin 
 test:=true; 
 if cap=nil then test:=false 
 else  
     begin 
     p:=cap; 
     cap:=cap^.leg; 
     atribuie(d,p^.inf); 
     dispose(p); 
     end; 
 end; 
 
c) Lista reprezentată prin structură dinamică dublu înlănţuită 
 
procedure elimina_prima3(var cap:lista;var d:tip_informatie;var 
test:boolean); 
var p:lista; 
 begin 
 test:=true; 
 if cap=nil then test:=false 
 else  
     begin 
     p:=cap; 
     cap:=cap^.dr; 
     cap^.st:=nil; 
     atribuie(d,p^.inf); 
     dispose(p); 
     end; 
 end; 

2.4  Liste circulare 
 
 

În anumite cazuri, este preferabilă renunţarea la structura de tip linear a 
listelor şi utilizarea unei legături de la ultima componentă către capul listei, rezultând 
ceea ce se numeşte listă circulară (figura 2.2). 

Avantajul utilizării acestui tip de structură este posibilitatea de accesare 
dintr-un element al listei a oricărui alt element. În continuare sunt prezentate module 
pentru realizarea unor operaţii de bază în lucrul cu liste circulare. 

 
type 
     tip_informatie=string; 
     clista=^nod; 
     nod=record 
     inf:tip_informatie; 
     leg:clista; 
     end; 

procedure atribuie(var d:tip_informatie;s:tip_informatie); 
begin 
d:=s; 
end; 
 
procedure inserare_la_inceput(var 
ultim:clista;d:tip_informatie;var test:boolean); 
var 
   p:clista; 
   cap:clista; 
begin 
     test:=true; 
     if MaxAvail<sizeof(nod) then test:=false 
     else 
         begin 
         new(p); 
         atribuie(p^.inf,d); 
         if ultim=nil then 
            begin 
            ultim:=p; 
            ultim^.leg:=ultim; 
            end 
         else 
             begin 
             cap:=ultim^.leg; 
             p^.leg:=cap; 
             ultim^.leg:=p; 
             end; 
         end; 
end; 
 
procedure inserare_la_sfarsit(var 
ultim:clista;d:tip_informatie;var test:boolean); 
var 
   p:clista; 
   cap:clista; 
begin 
     test:=true; 
     if MaxAvail<sizeof(nod) then test:=false 
     else 
         begin 
         new(p); 
         atribuie(p^.inf,d); 
         if ultim=nil then 
            begin 
            ultim:=p; 
            ultim^.leg:=ultim; 
            end 
         else 
             begin 
             cap:=ultim^.leg; 
             p^.leg:=cap; 
             ultim^.leg:=p; 
             ultim:=p; 
             end; 
         end; 
end; 
 
procedure elimina_la_inceput(var ultim:clista;var 
d:tip_informatie;var test:boolean); 
var 
   cap:clista; 

begin 
     test:=true; 
     if  ultim=nil then test:=false 
     else 
         begin 
         cap:=ultim^.leg; 
         atribuie(d,cap^.inf); 
         if cap=ultim then 
            begin 
            dispose(ultim); 
            ultim:=nil; 
            end 
         else 
             begin 
             ultim^.leg:=cap^.leg; 
             dispose(cap); 
             end; 
         end; 
end; 
 
procedure elimina_la_sfarsit(var ultim:clista;var 
d:tip_informatie;var test:boolean); 
var 
   p:clista; 
   cap:clista; 
begin 
     test:=true; 
     if  ultim=nil then test:=false 
     else 
         begin 
         cap:=ultim^.leg; 
         atribuie(d,ultim^.inf); 
         if cap=ultim then 
            begin 
            dispose(ultim); 
            ultim:=nil; 
            end 
         else 
             begin 
             p:=cap; 
             while(p^.leg<>ultim) do p:=p^.leg; 
             p^.leg:=cap; 
             dispose(ultim); 
             ultim:=p 
             end; 
         end; 
end; 
procedure afisare(ultim:clista); 
var p,cap:clista; 
begin 
if ultim=nil then writeln('Lista vida') 
else 
    begin 
    cap:=ultim^.leg; 
    p:=cap; 
    repeat 
          write(p^.inf,' '); 
          p:=p^.leg; 
    until p=cap; 
    writeln; 
    end; 
end; 

2.5  Stive şi cozi 
 
 

Accesul la informaţia memorată într-o listă lineară pentru efectuarea 
operaţiilor de inserare şi eliminare este permis la oricare dintre componentele 
colecţiei. Aşa după cum a rezultat în §2.3, identificarea poziţiei în care trebuie 
efectuată inserarea/eliminarea presupune iniţierea unei secvenţe de operaţii de 
căutare, ceea ce determină creşterea timpului de lucru. Pe de altă parte, o mulţime de 
aplicaţii pot fi modelate utilizând liste lineare în care introducerea şi, respectiv, 
eliminarea informaţiilor sunt permise numai la capete. Astfel au fost definite tipurile 
de listă  stivă şi coadă, care impun un tip de organizare a aplicării operaţiilor de 
inserare şi eliminare.  

Stiva 
Se numeşte stivă o listă organizată astfel încât operaţiile de inserare şi 

eliminare sunt permise numai la prima componentă. Acest mod de organizare 
corespunde unei gestiuni LIFO (Last In First Out) a informaţiei stocate. 

Modelul corespunde unei stive de cărţi. Adăugarea unei noi cărţi în stivă se 
face deasupra primei cărţi, iar extragerea este posibilă numai pentru prima carte. 

Operaţiile de inserare şi eliminare într-o stivă pot fi descrise prin intermediul 
procedurilor inserare_la_inceput1, inserare_la_inceput2, elimina_prima1, 
elimina_prima2. Operaţiile elementare pentru gestiunea informaţiei memorate într-o 
stivă sunt:  

- push(S,d,test) – inserarea informaţiei d în stiva  S; 
- pop(d,S,test) – preluarea cu eliminare a informaţiei memorate în prima 

celulă a stivei S; 
- top(d,S,test) – preluarea fără eliminare a informaţiei deţinute de prima 

componentă a stivei S. 
Parametrul test returnează true dacă operaţia este posibilă, altfel false. 
 
Coada 
Se numeşte coadă o listă organizată astfel încât operaţia de inserare este 

permisă la ultima componentă, iar operaţia de eliminare este permisă numai la prima 
componentă. Acest mod de organizare corespunde unei gestiuni FIFO (First In First 
Out) a informaţiei stocate. 

Modelul corespunde unei cozi de aşteptare la un magazin. O nouă persoană 
se aşază la coadă după ultimul cumpărător, iar persoana care îşi achită nota de plată 
(primul cumpărător) părăseşte coada. 

Implementarea unei liste coadă poate fi efectuată atât printr-o structură 
statică (masiv unidimensional), cât şi printr-o structură dinamică de tip listă. În 
scopul eficientizării operaţiilor de inserare/extragere, în cazul implementării cozilor 
prin structuri dinamice lineare, este necesară utilizarea a două informaţii: adresa 

primei componente şi adresa ultimei componente. Aceste informaţii pot fi menţinute 
explicit prin utilizarea a doi pointeri sau prin utilizarea unui pointer şi a unei structuri 
de listă circulară.  

1. Operaţiile de inserare şi eliminare pentru listă coadă reprezentată static 
sunt descrise de procedurile inserare_la_sfarsit1 şi eliminare_la_inceput prezentate 
în cadrul secţiunii §2.3.  

2. Reprezentarea unei liste coadă printr-o structură dinamică circulară 
utilizează o singură variabilă adresă, ultim, pentru referirea ultimei componente a 
listei. Operaţiile de inserare şi eliminare sunt descrise de procedurile inserare_coada 
şi eliminare_coada. 

 
procedure inserare_coada(var ultim:lista; d:tip_informatie; var 
test:boolean); 
var p:lista; 
begin 
 test:=true; 
 if MaxAvail<sizeof(nod) then test:=false 
 else 
  begin 
  new(p); 
  atribuie(p^.inf,d); 
  if ultim=nil then 
   begin 
   ultim:=p; 
   ultim^.leg:=ultim; 
   end 

else  
begin 
p^.leg:=ultim^.leg; 
ultim^.leg:=p; 
ultim:=p; 
end; 

  end; 
 end; 
procedure elimina_coada(var ultim:lista; var d:tip_informatie; 
var test:boolean); 
var prim:lista; 
begin 
 test:=true; 
 if ultim=nil then test:=false 
 else 
  begin 
  prim:=ultim^.leg; 
  atribuie(d,prim^.inf); 
  if ultim^.leg=ultim then  
   begin 
   dispose(ultim); 
   ultim:=nil; 
   end 
  else  

begin 
ultim^.leg:=prim^.leg; 
dispose(prim); 
end; 

  end; 
end; 

